

CS 457/557: Functional Languages

Equational Reasoning: Algebra of Programming

Mark P Jones
Portland State University

1


What Makes a Good Program?

- ◆ Performance?
- ◆ Code size?
- ◆ Maintainability?
- ◆ Above all else, correctness!
- ◆ But what does that mean? How can it be established?

2


Testing:

- ◆ Tests confirm expectations about the way things work
- ◆ If you drop a weight ...
- ◆ ... onto an egg ...
- ◆ ... Scrambled Egg!




Testing:

- ◆ Suppose it's our job to **protect** eggs from falling weights ...
- ◆ We might design an EP (Egg Protector™) to accomplish this ...
- ◆ Then we test again ...
- ◆ Hooray! The egg is safe! 😊





Generalizing from Tests:

- ◆ "The EP will protect an egg from a falling weight"
- ◆ Scrambled egg, and a crushed EP 😞
How embarrassing ...
- ◆ It can be **dangerous** to generalize from the results of testing!



Refining the claim:

- ◆ Think back to our test:
- ◆ "The EP will protect an egg from a falling weight **of at most 1kg**"
- ◆ This isn't such a general statement ...
- ◆ ... but it describes the EP's properties more accurately




More Tests:

- ◆ “The EP will protect an egg from a falling weight of at most 1kg”



- ◆ Oops, another embarrassing oversight!



Refining the EP Design:

- ◆ “The EP will protect an egg from a falling weight of at most 1kg”



Refining the EP Design:

- ◆ “The EP 2.0 will protect an egg from a falling weight of at most 1kg”




- ◆ We had to change the design of the EP ...

- ◆ But our egg is safe again!



Or is it?

- ◆ We’d like the EP to protect **any** egg ...



Or is it?

- ◆ We’d like the EP to protect **any** egg ...

- ◆ ... from **any** weight ...



General Observations:


- ◆ Testing helps us to find (and then avoid):
 - bugs in the things that we build
 - bugs in the claims that we make about them
- ◆ Testing and Development working together ...
- ◆ But ...



Testing has Limits:

◆ "testing can be used to show the presence of bugs, but never to show their absence" [Edsger Dijkstra, 1969]

◆ To be **absolutely certain** that the EP 2.0 will protect **any egg** from **any weight** under 1kg, we will need to **prove it**.



Equational Reasoning:

- ◆ Functional Languages are Good for Equational Reasoning (Gofer!)
- ◆ Much of what follows is inspired by the work of Richard Bird
- ◆ **Goal:** to prove laws of the form $e_1 = e_2$ relating program fragments e_1 and e_2
- ◆ **Goal:** to calculate/synthesize efficient definitions of functions from clear, high-level specifications

14

Laws of Numbers:

If n is a natural number, then either:

$n = 0$; or

$n = 1 + m$ for some (smaller) natural m

Functions on natural numbers:

$0 + n = n$

$(1+m) + n = 1 + (m + n)$

Does this look at all familiar?

15

+ is associative:

$\forall n. \forall p. \forall q. (n + p) + q = n + (p + q)$

If $n = 0$, then

$(n + p) + q$

$= (0 + p) + q$

(because $n = 0$)

$= p + q$

(definition of +)

$= 0 + (p + q)$

(definition of +)

16

+ is associative:

$\forall n. \forall p. \forall q. (n + p) + q = n + (p + q)$

If $n = (1+m)$, then

$(n + p) + q$

$= ((1 + m) + p) + q$ (because $n=1+m$)

$= (1 + (m + p)) + q$ (definition of +)

$= 1 + ((m + p) + q)$ (definition of +)

$= 1 + (m + (p + q))$ (induction)

$= (1 + m) + (p + q)$ (definition of +)

$= n + (p + q)$ (definition of +)

17

+ is associative:

We've shown:

- The property holds for $n = 0$
- If the property holds for $n = m$, then it holds for $n = (1 + m)$
- So it holds for $n = 1$
- And for $n = 2$
- And for $n = 3$
- ...

In fact, we've shown that it holds for all n :

$\forall n. \forall p. \forall q. (n + p) + q = n + (p + q)$

18

Laws of Numbers:

If n is a natural number, then either:

$n = \text{Zero}$; or

$n = \text{Succ } m$ for some (smaller) natural m

`data Nat = Zero | Succ Nat`

Functions on natural numbers:

`add Zero n = n`

`add (Succ m) n = Succ (add m n)`

19

add is associative:

$\forall n. \forall p. \forall q. \text{add } (\text{add } n \text{ } p) \text{ } q = \text{add } n \text{ } (\text{add } p \text{ } q)$

If $n = \text{Zero}$, then

`add (add n p) q`

`= add (add Zero p) q` (because $n = \text{Zero}$)

`= add p q` (definition of `add`)

`= add Zero (add p q)` (definition of `add`)

20

add is associative:

$\forall n. \forall p. \forall q. \text{add } (\text{add } n \text{ } p) \text{ } q = \text{add } n \text{ } (\text{add } p \text{ } q)$

If $n = \text{Succ } m$, then

`add (add n p) q`

`= add (add (Succ m) p) q` (because $n = 1 + m$)

`= add (Succ (add m p)) q` (definition of `+`)

`= Succ (add (add m p) q)` (definition of `+`)

`= Succ (add m (add p q))` (induction)

`= add (Succ m) (add p q)` (definition of `+`)

`= add n (add p q)` (definition of `+`)

21

add is associative:

We've shown:

- The property holds for $n = \text{Zero}$
- If the property holds for $n = m$, then it holds for $n = \text{Succ } m$
- So it holds for $n = \text{Succ Zero}$
- And for $n = \text{Succ (Succ Zero)}$
- And for $n = \text{Succ (Succ (Succ Zero))}$
- ...

In fact, we've shown that it holds for all n :

$\forall n. \forall p. \forall q. \text{add } (\text{add } n \text{ } p) \text{ } q = \text{add } n \text{ } (\text{add } p \text{ } q)$

22

Laws in Haskell:

We can apply these same ideas to many other Haskell datatypes, not just numbers

Algebra for programs:

- ◆ Break into cases (no junk, no confusion)
- ◆ Induction (recursion)
- ◆ Equational reasoning

23

Where do Laws come From?

Laws typically arise in one of three ways:

- ◆ From function definitions (with care)
`(x:xs) ++ ys = x : (xs ++ ys)`
- ◆ From previously established laws
`map f . map g = map (f . g)`
- ◆ From specifications of new functions
`sumSquares n = sum (map square [1..n])`

24

Referential Transparency:

- ◆ The ability to replace equals with equals
 - If $e_1=e_2$, then $\dots e_1\dots = \dots e_2\dots$
- ◆ The inability to observe sharing
 - **let** $x = e$ **in** $(x,x) = (e, e)$
 - **let** $x = \text{print } 1$ **in** $(x,x) = (\text{print } 1, \text{print } 1)$

25

Tools:

- ◆ Extensionality:
 - $f = g \Leftrightarrow \forall x. f\ x = g\ x$
- ◆ Simple substitution/instantiation:
 - From $(f . g)\ x = f\ (g\ x)$, we can infer that $((1+) . (2*))\ n = 1 + 2*n$

26

... continued:

- ◆ Case analysis:
 - If $xs :: [a]$, then $xs = []$, or $xs = (y:ys)$ for some y and ys , or $xs = \perp$
 - If $b :: \text{Bool}$, then $b = \text{False}$, $b = \text{True}$, or $b = \perp$
- ◆ Induction:
 - If property $P(xs)$ holds for $xs = []$ and for $xs = \perp$, and for $(y:ys)$ whenever it holds for ys , then $P(xs)$ holds for all lists xs .

27

Introducing Bottom, \perp :

- ◆ We treat every type in Haskell as having a special element called bottom, written \perp
- ◆ \perp represents the value produced by expressions that fail to terminate properly
 - Non-termination
 - Error (e.g., missing pattern matching case)
 - Explicit call of `error "... message ..."`
- ◆ Called "bottom" because it has the least amount of information of any value

28

Strictness:

- ◆ We say that a function is strict if it is guaranteed to evaluate its argument.
- ◆ Another way to say this: f is strict if, and only if $f\ \perp = \perp$
- ◆ Examples:
 - $(1+)$ and `not` are both strict
 - `(&&)` and `(||)` are strict in their left arguments, but not in their right
 - `map` is strict in its list argument (but not the function)

29

Example:

- ◆ Suppose we specify:
 - $f :: [\text{Int}] \rightarrow [\text{Int}]$
 - $f = \text{map } (1+)$
- ◆ Now we can calculate:
 - $f\ []$
 - $= \{ \text{by definition of } f \}$
 - $\text{map } (1+)\ []$
 - $= \{ \text{by definition of } \text{map} \}$
 - $[]$

30

... continued:

- ◆ We can also calculate:

```
f (x:xs)
= { by definition of f }
  map (1+) (x:xs)
= { by definition of map }
  (1+x) : map (1+) xs
= { by definition of f }
  (1 + x) : f xs
```

- ◆ Thus we have derived:

```
f      :: [Int] -> [Int]
f []   = []
f (x:xs) = (1+x) : f xs
```

31

Associativity of (++):

Claim: $xs++(ys++zs) = (xs++ys)++zs$, for all xs , ys , and zs

Proof by induction on xs :

```
Base case: xs = []
[] ++ (ys ++ zs)
= { by definition of ++ }
  ys ++ zs
= { by definition of ++ }
  ([] ++ ys) ++ zs
```

32

... continued:

Base case: $xs = \perp$

```
lhs:  \perp ++ (ys ++ zs)
      = { ++ is strict in its first argument }
        \perp
```

```
rhs:  (\perp ++ ys) ++ zs
      = { ++ is strict in its first argument }
        \perp ++ zs
      = { ++ is strict in its first argument }
        \perp
```

33

... continued:

Inductive case: $(x:xs)$

```
(x:xs) ++ (ys ++ zs)
= { by definition of ++ }
  x : (xs ++ (ys ++ zs))
= { by induction }
  x : ((xs ++ ys) ++ zs)
= { by definition of ++ }
  (x : (xs ++ ys)) ++ zs
= { by definition of ++ }
  ((x:xs) ++ ys) ++ zs
```

34

Fold Right:

A function from the prelude:

```
foldr :: (a -> b -> b) -> b -> [a] -> b
foldr (\oplus) e [x0,x1,x2] = x0 \oplus (x1 \oplus (x2 \oplus e))
```

Examples:

```
and  = foldr (&&) True
concat = foldr (++) []
```

Definition:

```
foldr f e [] = e
foldr f e (x:xs) = f x (foldr f e xs)
```

35

Fold Left:

A function from the prelude:

```
foldl :: (a -> b -> a) -> a -> [b] -> a
foldl (\oplus) e [x0,x1,x2] = ((e \oplus x0) \oplus x1) \oplus x2
```

Examples:

```
sum  = foldl (+) 0
product = foldl (*) 1
```

Definition:

```
foldl f e [] = e
foldl f e (x:xs) = foldl f (f e x) xs
```

36

Scan Left:

A function from the prelude:

```
scanl :: (a -> b -> a) -> a -> [b] -> [a]
scanl (⊕) e [x0, x1, x2]
= [ e, e⊕x0, (e⊕x0)⊕x1, ((e⊕x0)⊕x1)⊕x2 ]
```

Specification:

```
scanl f e = map (foldl f e) . inits
```

```
inits [] = [[]]
```

```
inits (x:xs) = [] : map (x:) (inits xs)
```

37

Calculating scanl:

It is easy to derive $\text{scanl } f \ e \ [] = [e]$

For non empty lists:

```
scanl f e (x:xs)
= map (foldl f e) (inits (x:xs))
= map (foldl f e) ([] : map (x:) (inits xs))
= foldl f e [] : map (foldl f e) (map (x:) (inits xs))
= foldl f e [] : map (foldl f e . (x:)) (inits xs)
= e : map (foldl f (f e x)) (inits xs)
= e : scanl f (f e x) xs
```

38

Comparison:

◆ Specification:

```
scanl f e = map (foldl f e) . inits
```

◆ Definition:

```
scanl f e [] = [e]
```

```
scanl f e (x:xs) = e : scanl f (f e x) xs
```

◆ The specification requires $O(n^2)$ applications of f on a list of length n while the definition uses only n applications for a list of the same length.

◆ But, in terms of the results that we obtain, we know that the two versions are equal!

39

Scan Right:

A dual of scanl:

```
scanr :: (a -> b -> b) -> b -> [a] -> [b]
scanr f e = map (foldr f e) . tails
```

```
scanr (⊕) e [x0, x1, x2]
= [x0⊕(x1⊕(x2⊕ e)), x1⊕(x2⊕ e), x2⊕ e, e]
```

More efficient version:

```
scanr f e [] = [e]
```

```
scanr f e (x:xs) = f x (head ys) : ys
```

where $ys = \text{scanr } f \ e \ xs$

40

Maximum Segment Sum:

◆ Given a sequence of numbers, find the subsegment whose sum is largest:

- Example: maximal subsegment sum for the list $[-1, 2, -3, 5, -2, 1, 3, -2, -2, -3, 6]$ is 7 (for the segment $[5, -2, 1, 3]$)

◆ Simple solution:

```
mss :: [Int] -> Int
```

```
mss = maximum . map sum . segs
```

where $\text{segs} = \text{concat} . \text{map inits} . \text{tails}$

◆ Not a great performer ... $O(n^3)$

41

Calculate!

```
mss
```

```
= {definition of mss}
```

```
maximum . map sum . segs
```

42

Calculate!

```
mss
= {definition of segs}
  maximum . map sum . concat . map inits . tails
```

43

Calculate!

```
mss
= {using map f . concat = concat . map (map f) }
  maximum . concat . map (map sum) . map inits . tails
```

```
(map f . concat) [xs1, xs2, xs3]
= map f (xs1 ++ xs2 ++ xs3)
= map f xs1 ++ map f xs2 ++ map f xs3
```

```
(concat . map (map f)) [xs1, xs2, xs3]
= concat [map f xs1, map f xs2, map f xs3]
= map f xs1 ++ map f xs2 ++ map f xs3
```

44

Calculate!

```
mss
= { using map f . map g = map (f . g) }
  maximum . concat . map (map sum . inits) . tails
```

```
(map f . map g) [ x1, x2, x3 ]
= map f [ g x1, g x2, g x3 ]
= [ f (g x1), f (g x2), f (g x3) ]
```

```
map (f . g) [ x1, x2, x3 ]
= [ (f . g) x1, (f . g) x2, (f . g) x3 ]
= [ f (g x1), f (g x2), f (g x3) ]
```

45

Calculate!

```
mss
= { the "bookkeeping law" }
  maximum . map maximum . map (map sum . inits) . tails
```

```
maximum . concat
= maximum . map maximum
```

General form:

```
foldr f a . concat = foldr f a . map (foldr f a)
if f is associative with unit a
```

46

Calculate!

```
mss
= { Definition of scanl }
  maximum . map maximum . map (scanl (+) 0) . tails
```

```
Definition:
scanl f e = map (foldl f e) . inits
```

47

Calculate!

```
mss
= {using map f . map g = map (f . g) }
  maximum . map (maximum . scanl (+) 0) . tails
```

```
map f . map g = map (f . g)
                (again ...)
```

48

Calculate!

```
mss
= { fold-scan fusion }
maximum . map (foldr f 0) . tails
  where f x y = max 0 (x + y)
```

We can prove that:
maximum . scanl (+) 0 = foldr f 0
(A special case of a general property called "Fold-scan fusion")

49

Calculate!

```
mss
= { definition of scanr }
maximum . scanr f 0
  where f x y = max 0 (x + y)
```

Definition of scanr:
scanr f e = map (foldr f e) . tails

A simple, linear time algorithm,
courtesy of equational reasoning!

50

Calculate!

```
mss
= { definition of scanr }
maximum . scanr f 0
  where f x y = max 0 (x + y)
```

Remember:
scanr (\oplus) e [x₀, x₁, x₂]
= [x₀⊕(x₁⊕(x₂⊕ e)),
x₁⊕(x₂⊕ e),
x₂⊕ e,
e]

```
mss xs = loop 0 0 (reverse xs)
where
  loop m v [] = m
  loop m v (x:xs) = let y = max 0 (x+v)
                    in loop (max m y) y xs
```

This version of the definition is
not very intuitive ... but we know
by construction that it is correct!

51

A Quick Check:

Just to be sure, let's load these definitions in replugs
and quickly check to see if they are

```
Main> quickCheck
OK, passed
```

To Be Continued ...

Hmm, now that looks like another useful tool,
doesn't it ...

52

Summary:

- ◆ The ability to reason about code is essential if you care about its behavior (for example, in safety or security critical applications)
- ◆ Compilers rely on equivalences between program fragments to justify/validate some optimizations
- ◆ Functional Languages are Good for Equational Reasoning
- ◆ Referential transparency/lack of side effects makes reasoning more tractable
- ◆ It helps to build up a collection of laws and results that you can draw on in program verification or synthesis!

53